OneCo Implementation Showcase

In $\mathcal{U}(T)\otimes\mathcal{U}(H)$ conventions.

Generalities.

```
Simp[ε_] := Expand[ε];
S[ε_] :=
  ε /. (λ_β | λ_a | λ_δβ | λ_δa | λ_δaa) ⧴ MapAt[Simp, λ, 1];
AutoCollecting[λ_] := (λ /: λ[0, ___] = 0;
   λ /: λ[f_, r___] + λ[g_, r___] := λ[Simp[f + g], r];
   λ /: g_ * λ[f_, r___] := λ[Simp[g f], r]);
AutoCollecting /@ {β, a, δβ, δa, δaa};
UU /: UU[x_] + UU[y_] := UU[x + y];
UU /: a_ * UU[x_] := UU[Expand[a x]];
UU /: D[u_UU, vs__] :=
  S[u /. (λ_β | λ_a | λ_δβ | λ_δa | λ_δaa) ⧴
     MapAt[D[#, vs] &, λ, 1]];
b_ç = 1; ca[f_, i_, j_, k_] := δaa[f, ç, i, j, k];
γ[f_, j_, k_] := δa[f, j, k] - δa[b_j f, ç, k];
γa[f_, j_, k_, l_, m_] :=
  δaa[f, j, k, l, m] - ca[b_j f, k, l, m];
Kδ /: Kδ_is__ := KroneckerDelta[1, Length[Union[{is}]]];
```

δaa relations.

```
i_ ≤ j_ := OrderedQ[{i, j}]; i_ < j_ := ! OrderedQ[{j, i}];
S[UU[ε_]] := UU[S[ε /. δaa[f_, i_, j_, k_, l_] ⧴ Which[
      k === ç, δaa[f, ç, l, i, j] + Kδ_jl γ[f, i, j],
      (i === ç) ⋁ (i ≤ k ⋀ j ≤ l), δaa[f, i, j, k, l],
      k < i ⋀ j < l, δaa[f, k, j, i, l] + ca[-f b_i, l, k, j] +
       ca[f b_i, j, k, l] + ca[-f b_k, j, i, l] +
       ca[f b_k, l, i, j],
      k < i ⋀ j === l, δa[-f b_i, k, j] + δa[f b_k, i, j] +
       δaa[f, k, j, i, j],
      i ≤ k ⋀ l < j, δaa[f, i, l, k, j] + ca[-f b_i, l, k, j] +
       ca[f b_i, j, k, l] + ca[-f b_k, j, i, l] +
       ca[f b_k, l, i, j],
      k < i ⋀ l < j, δaa[f, k, l, i, j]
     ]]];
```

Definition of *tm*.

```
UU[ε_] // tm[x_, y_, z_] := (rr = Replace[x | y → z];
   S[UU[Expand[ε /. {
         a[f_, x, j_] ⧴ a[f, z, j] + γ[∂_{b_y} f, z, j],
         a[f_, y, j_] ⧴ a[f, z, j],
         δa[f_, x | y, j_] ⧴ δa[f, z, j],
         δaa[f_, i_, j_, k_, l_] ⧴
          δaa[f, rr@i, j, rr@k, l]
        } /. b_{x|y} → b_z]]]);
```

Definition of *hm*.

```
UU[ε_] // hm[x_, y_, z_] := (rr = Replace[x | y → z];
   S[UU[Expand[ε /. {
        a[f_, i_, x | y] ⧴ a[f, i, z],
        δa[f_, i_, x | y] ⧴ δa[f, i, z],
        δaa[f_, i_, y, k_, x] ⧴ δaa[f, k, z, i, z],
        δaa[f_, i_, j_, k_, l_] ⧴ δaa[f, i, rr@j, k, rr@l]
       }]]]);
```

Definition of *hts*.

```
UU[ε_] // hts[y_, x_] := S[UU[Expand[ε /. {
       a[f_, i_, j_] ⧴ a[f, i, j] - Kδ_jy γ[∂_{b_x} f, i, y] -
         Kδ_ix Kδ_jy (β[f b_x] - δa[f, ç, y] - δβ[b_x ∂_{b_x} f]),
       δa[f_, x, y] ⧴ δa[f, x, y] - δβ[f b_x],
       δaa[f_, i_, j_, k_, l_] ⧴
        δaa[f, i, j, k, l] + Kδ_ix Kδ_jy δa[-b_x f, k, l] +
         Kδ_ix Kδ_ly (δa[b_k f, x, j] - δa[b_x f, k, j]) +
         Kδ_kx Kδ_jy (δa[b_i f, x, l] - δa[b_x f, i, l]) +
         Kδ_kx Kδ_ly δa[-b_x f, i, j] - Kδ_ix Kδ_jly δβ[b_x b_k f] +
         2 Kδ_xik Kδ_yjl δβ[b_x b_x f]
      }]]];
```

Definition of *dm*.

```
dm[x_, y_, z_][ε_] :=
  ε // hts[x, y] // tm[x, y, z] // hm[x, y, z]
```

Renaming operations.

```
tσ[x_List, y_List][ε_] := (rr = Replace[Thread[x → y]];
   S[ε /. b_i_ ⧴ b_rr@i /. {
        a[f_, i_, j_] ⧴ a[f, rr@i, j],
        δa[f_, i_, j_] ⧴ δa[f, rr@i, j],
        δaa[f_, i_, j_, k_, l_] ⧴ δaa[f, rr@i, j, rr@k, l]
       }]);
tσ[x_, y_][ε_] := tσ[{x}, {y}][ε];
hσ[x_List, y_List][ε_] := (rr = Replace[Thread[x → y]];
   S[ε /. {
        a[f_, i_, j_] ⧴ a[f, i, rr@j],
        δa[f_, i_, j_] ⧴ δa[f, i, rr@j],
        δaa[f_, i_, j_, k_, l_] ⧴ δaa[f, i, rr@j, k, rr@l]
       }]);
hσ[x_, y_][ε_] := hσ[{x}, {y}][ε];
dσ[x_, y_][ε_] := ε // tσ[x, y] // hσ[x, y];
```

Definition of *tb*.

```
tb[x_][UU[L_], UU[R_]] :=
  S[UU[Expand[Distribute[pp[L, R]] /. {
        pp[0, _] → 0, pp[_, 0] → 0,
        pp[_β | _δβ | _δa | _δaa, _β | _δβ | _δa | _δaa] → 0,
        pp[u_β | u_δβ | u_δa | u_δaa, v_a] ⧴ -pp[v, u]
       } /. {
       pp[a[f_, x, j_], u_] ⧴ (u /. {
            β[g_] ⧴ γ[f ∂_{b_x} g, x, j],
            a[g_, k_, l_] ⧴ γa[f ∂_{b_x} g, x, j, k, l] +
              Kδ_xk (-γa[g ∂_{b_k} f, k, l, x, j] +
                 ca[f g, l, x, j] - ca[f g, j, k, l]),
            _ → 0
           }),
       pp[a[f_, j_, k_], a[g_, x, l_]] /; j =!= x ⧴
        -γa[g ∂_{b_x} f, x, l, j, k],
       pp[_, _] → 0
      }]]];
```

Definition of *thb*.

```
thb[x_, y_][UU[L_], UU[R_]] :=
  S[UU[Expand[Distribute[pp[L, R]] /. {
      pp[0, _] → 0, pp[_, 0] → 0,
      pp[_β | _δβ | _δa | _δaa, _β | _δβ | _δa | _δaa] → 0,
      pp[_a, _β | _δβ] → 0,
      pp[β[f_], a[g_, i_, j_]] :→ Kδ_yj γ[g ∂_bx f, i, y],
      pp[a[f_, i_, j_], a[g_, k_, l_]] :→ Kδ_yl (
         γa[g ∂_bx f, k, l, i, j] + Kδ_xi (
           γ[-b_k g ∂_bx f, i, j] + δa[b_k g ∂_bx f, i, j] -
            δa[b_i g ∂_bx f, k, j] - a[b_k f g, i, j] +
            a[b_i f g, k, j] + ca[f g, j, k, l] -
            ca[f g, l, k, j])),
      pp[a[f_, i_, j_], δa[g_, k_, l_]] :→
        Kδ_xi Kδ_yl (-δa[b_k f g, i, j] + δa[b_i f g, k, j]),
      pp[a[f_, i_, j_], δaa[g_, k_, l_, m_, n_]] :→ Kδ_xi (
         Kδ_yl (-δaa[b_k f g, i, j, m, n] +
            δaa[b_i f g, k, j, m, n]) +
          Kδ_yn (-δaa[b_m f g, k, l, i, j] +
            δaa[b_i f g, k, l, m, j]) +
          Kδ_yln (δa[b_k b_m f g, k, j] - δa[b_k b_m f g, x, j])),
      pp[_δβ, _a] → 0,
      pp[δa[f_, i_, j_], a[g_, k_, l_]] :→
        Kδ_xi Kδ_yl (-δa[b_k f g, i, j] + δa[b_i f g, k, j]),
      pp[δaa[f_, i_, j_, m_, n_], a[g_, k_, l_]] :→
        Kδ_xi Kδ_yl (-δaa[b_k f g, i, j, m, n] +
           δaa[b_i f g, k, j, m, n]) +
         Kδ_xm Kδ_yl (-δaa[b_k f g, i, j, m, n] +
           δaa[b_m f g, i, j, k, n])
    }]]];
htb[x_, y_][L_UU, R_UU] := -thb[y, x][R, L];
```

aY v.s Ya ?

Definition of *hb*.

```
hb[y_][UU[L_], UU[R_]] :=
  S[UU[Expand[Distribute[pp[L, R]] /. {
       pp[0, _] → 0, pp[_, 0] → 0,
       pp[_β | _δβ, _] → 0,
       pp[_, _β | _δβ] → 0,
       pp[_δa | _δaa, _δa | _δaa] → 0,
       pp[u_δa | u_δaa, v_a] :→ -pp[v, u]
     } /. {
       pp[a[f_, i_, y], u_] :→ (u /. {
          a[g_, j_, k_] :→
           Kδ_yk (a[b_j f g, i, y] - a[b_i f g, j, k]),
          δa[g_, j_, k_] :→
           Kδ_yk (δa[b_j f g, i, y] - δa[b_i f g, j, k]),
          δaa[g_, j_, k_, l_, m_] :→
           Kδ_yk (δaa[b_j f g, i, y, l, m] -
              δaa[b_i f g, j, k, l, m]) +
            Kδ_ym (δaa[b_l f g, j, k, i, y] -
              δaa[b_i f g, j, k, l, m])
         }),
       _pp → 0
     }]]];
```

Definition of *db*.

Using $h_1h_2t_1t_2 \to h_1h_2t_1t_2 \to h_1h_2t_2t_1 \to h_2h_1t_2t_1 \to h_2h_1t_2t_1$:

```
db[x_][u_UU, v_UU] := Module[{t, h}, Plus[
    htb[x, x][u // tσ[x, t], v // hσ[x, h]] // tm[t, x, x] //
     hm[x, h, x],
    tb[x][u, v // hσ[x, h]] // hm[x, h, x],
    hb[x][u, v // tσ[x, t]] // tm[t, x, x],
    thb[x, x][u // hσ[x, h], v // tσ[x, t]] //
      tm[t, x, x] // hm[x, h, x] ]];
```

The bracket.

```
bb[S_List] := Module[{w, bar, t, n = 0, i, k},
   w = #2 // dσ[S, bar /@ S];
   Sum[t = db[S⟦k⟧][#1, w // dσ[bar[S⟦k⟧], S⟦k⟧]];
    Do[t = t // dm[bar[S⟦i⟧], S⟦i⟧, S⟦i⟧], {i, 1, k - 1}];
    Do[t = t // dm[S⟦i⟧, bar[S⟦i⟧], S⟦i⟧],
     {i, k + 1, Length@S}];
   t, {k, Length@S}] ] &
bb[S___] := bb[{S}]
```

Recovery challenges.

```
AutoAd[bb[j, k], UU@a[1, j, k]][UU@a[1, 0, k]]
```

$$\mathrm{UU}\Big[a[e^{-b_j}, 0, k] + a\Big[\frac{(1-e^{-b_j})\, b_0}{b_j}, j, k\Big] + \delta a\Big[\frac{e^{-2b_j} b_0 (-1-e^{b_j}(-1+b_j))}{b_j^2}, j, k\Big] + \delta a\Big[\frac{e^{-2b_j}(1+e^{b_j}(-1+b_j))}{b_j}, 0, k\Big] + \delta aa\Big[\frac{e^{-2b_j}(-1-e^{b_j}(-1+b_j))}{b_j^2}, 0, k, j, k\Big] + \delta aa\Big[\frac{2e^{-b_j} b_0 (\mathrm{Sinh}[b_j]-b_j)}{b_j^2}, ç, k, j, k\Big] + \delta aa\Big[\frac{e^{-2b_j}(1+e^{b_j}(-1+b_j))}{b_j}, ç, k, 0, k\Big] + \delta aa\Big[\frac{2e^{-b_j} b_0 (-\mathrm{Sinh}[b_j]+b_j)}{b_j^3}, j, k, j, k\Big]\Big]$$

```
AutoAd[bb[j, k], UU@a[1, j, k]][UU@a[1, 0, j]]
```

$$\mathrm{UU}\Big[a[1, 0, j] + a[1-e^{-b_j}, 0, k] + a\Big[\frac{(-1+e^{-b_j})\, b_0}{b_j}, j, k\Big] + \delta a\Big[-1-e^{-b_j}+\frac{1-e^{-2b_j}}{b_j}, 0, k\Big] + \delta a\Big[b_0\Big(1+\frac{-1+e^{-b_j}}{b_j}\Big), ç, k\Big] + \delta a\Big[\frac{e^{-2b_j} b_0 (1+e^{b_j}(-1+b_j))}{b_j^2}, j, k\Big] + \delta aa\Big[\frac{2e^{-b_j} b_0 (\mathrm{Sinh}[b_j]-b_j)}{b_j^3}, j, k, j, k\Big] + \delta aa\Big[\frac{b_0 (1-e^{-b_j}-b_j)}{b_j^2}, ç, j, j, k\Big] + \delta aa\Big[\frac{1-e^{-b_j}}{b_j}, ç, k, 0, j\Big] + \delta aa\Big[\frac{-1+e^{-b_j}}{b_j}, ç, j, 0, k\Big] + \delta aa\Big[\frac{e^{-2b_j}(-1-e^{b_j}(-1+b_j))}{b_j}, ç, k, 0, k\Big] + \delta aa\Big[\frac{-1+e^{-b_j}+b_j}{b_j^2}, 0, j, j, k\Big] + \delta aa\Big[\frac{e^{-2b_j}(1+e^{2b_j}(-1+b_j)+e^{b_j} b_j)}{b_j^2}, 0, k, j, k\Big] + \delta aa\Big[-\frac{e^{-2b_j} b_0 (-1+e^{b_j}+e^{b_j}(-2+e^{b_j}) b_j)}{b_j^2}, ç, k, j, k\Big]\Big]$$

Definition of *ct*.

```
ct[s_] := ct[s, s]; ct[] = ct[0, 0];
ct[h_, t_][UU[L_], UU[R_]] := S[UU[Distribute[pp[L, R]] /. {
    pp[_β | _δβ, _] → 0,
    pp[a[f_, i_, h], β[g_]] :→ β[f b_i ((∂_{b_t} g) /. b_t → 0)],
    pp[a[f_, i_, h], a[g_, t, j_]] :→ a[f (g /. b_t → 0), i, j],
    pp[a[f_, i_, h], a[g_, j_, k_]] :→ a[f b_i ((∂_{b_t} g) /. b_t → 0), j, k],
    pp[a[f_, i_, h], δa[g_, t, j_]] :→ δa[f (g /. b_t → 0), i, j],
    pp[a[f_, i_, h], δa[g_, j_, k_]] :→ δa[f b_i ((∂_{b_t} g) /. b_t → 0), j, k],
    pp[a[f_, i_, h], δaa[g_, t, j_, t, k_]] → 0,
    pp[a[f_, i_, h], δaa[g_, t, j_, k_, l_]] :→ δaa[f (g /. b_t → 0), i, j, k, l],
    pp[a[f_, i_, h], δaa[g_, j_, k_, t, l_]] :→ δaa[f (g /. b_t → 0), j, k, i, l],
    pp[a[f_, i_, h], δaa[g_, j_, k_, l_, m_]] :→ δaa[f b_i ((∂_{b_t} g) /. b_t → 0), j, k, l, m],
    pp[a[__], _] → 0, pp[_δa | _δaa, _δβ | _δa | _δaa] → 0,
    pp[δa[f_, i_, h], β[g_]] :→ δβ[f b_i ((∂_{b_t} g) /. b_t → 0)],
    pp[δa[f_, i_, h], a[g_, t, j_]] :→ δa[f (g /. b_t → 0), i, j],
    pp[δa[f_, i_, h], a[g_, j_, k_]] :→ δa[f b_i ((∂_{b_t} g) /. b_t → 0), j, k],
    pp[_δa, _] → 0, pp[δaa[_, _, h, _, h], _] → 0,
    pp[δaa[f_, i_, h, j_, k_], β[g_]] :→ δa[f b_i ((∂_{b_t} g) /. b_t → 0), j, k],
    pp[δaa[f_, i_, h, j_, k_], a[g_, t, l_]] :→ δaa[f (g /. b_t → 0), i, l, j, k],
    pp[δaa[f_, i_, h, j_, k_], a[g_, l_, m_]] :→ δaa[f b_i ((∂_{b_t} g) /. b_t → 0), j, k, l, m],
    pp[δaa[f_, i_, j_, k_, h], β[g_]] :→ δa[f b_k ((∂_{b_t} g) /. b_t → 0), i, j],
    pp[δaa[f_, i_, j_, k_, h], a[g_, t, l_]] :→ δaa[f (g /. b_t → 0), i, j, k, l],
    pp[δaa[f_, i_, j_, k_, h], a[g_, l_, m_]] :→ δaa[f b_k ((∂_{b_t} g) /. b_t → 0), i, j, l, m],
    pp[_δaa, _] → 0 }]];
```

Global Generalities.

```
SnG[λ_, _][k_] := λ[k];
SnG[_, ω_][ω] := ω;
```

Exponentiating an arrow.

```
E_{j_,k_,t_} := SnG[⟨|
    j → UU[a[1, j, ∞] + δaa[t, ç, ∞, j, k] +
        δaa[(-1 + e^(-t b_j))/b_j, ç, k, j, ∞] +
        δaa[-(-1 + e^(-t b_j) + t b_j)/b_j^2, j, k, j, ∞]],
    k → UU[a[e^(t b_j), k, ∞] + a[-((-1 + e^(t b_j)) b_k)/b_j, j, ∞] +
        δa[((-1 + e^(t b_j) (1 - t b_j)) b_k)/b_j^2, j, ∞] +
        δa[((1 + e^(t b_j) (-1 + t b_j)) b_k)/b_j, ç, ∞] +
        δaa[(-1 + e^(t b_j) (1 - t b_j))/b_j, ç, k, k, ∞] +
        δaa[(1 + e^(t b_j) (-1 + t b_j))/b_j^2, j, k, k, ∞] +
        δaa[(b_j - e^(-t b_j) b_j + b_k + e^(t b_j) (-1 + t b_j) b_k)/b_j^2, ç, k, j, ∞] +
        δaa[(b_j + b_k + t b_j b_k - e^(t b_j) (b_j + b_k))/b_j^2, ç, ∞, j, k] +
        δaa[1/b_j^3
          e^(-t b_j) (b_j + e^(2 t b_j) (b_j + (2 - t b_j) b_k) -
            e^(t b_j) (2 b_k + b_j (2 + t b_k))), j, k, j, ∞]]|⟩,
  UU@a[t, j, k]];
E_{j_,k_} := E_{j,k,t};
```

Verifying the exponentiation.

```
{E_{1,2,t}[1], E_{1,2,t}[2]} /. t → 0
{UU[a[1, 1, ∞]], UU[a[1, 2, ∞]]}
{D[E_{1,2,t}[1], t], D[E_{1,2,t}[2], t]}
{UU[δaa[1, ç, ∞, 1, 2] +
    δaa[-e^(-t b_1), ç, 2, 1, ∞] + δaa[-1/b_1 + e^(-t b_1)/b_1, 1, 2, 1, ∞]],
 UU[a[e^(t b_1) b_1, 2, ∞] + a[-e^(t b_1) b_2, 1, ∞] +
    δa[-e^(t b_1) t b_2, 1, ∞] + δa[e^(t b_1) t b_1 b_2, ç, ∞] +
    δaa[e^(t b_1) t, 1, 2, 2, ∞] + δaa[-e^(t b_1) t b_1, ç, 2, 2, ∞] +
    δaa[e^(-t b_1) + e^(t b_1) t b_2, ç, 2, 1, ∞] +
    δaa[-e^(t b_1) + b_2/b_1 - e^(t b_1) b_2/b_1, ç, ∞, 1, 2] +
    δaa[-e^(-t b_1)/b_1 + e^(t b_1)/b_1 - b_2/b_1^2 + e^(t b_1) b_2/b_1^2 - e^(t b_1) t b_2/b_1, 1, 2, 1, ∞]]}

{D[E_{1,2,t}[1], t] - bb[1, 2][UU[a[1, 1, 2]], E_{1,2,t}[1]],
 D[E_{1,2,t}[2], t] - bb[1, 2][UU[a[1, 1, 2]], E_{1,2,t}[2]]}
{UU[0], UU[0]}
```